

AULA 4 – ESTRUTURAS DE REPETIÇÃO

@PROF.WALTERTRAVASSOS

WALTER.TRAVASSOS@SECTRAS.EDU.BR

Aula de Hoje

Estruturas de repetição:

- enquanto
- faça...enquanto
- para

Estruturas de Repetição

Uma estrutura de repetição é utilizada quando...

- Um trecho do algoritmo (lista de instruções) ou até mesmo o algoritmo inteiro precisa ser repetido.

O número de repetições poderá ser fixo ou estar relacionado a uma condição.

- **Fixo**, quando sabemos quantas vezes um determinado trecho de código será executado.
- **Relacionado a uma condição**, quando a repetição de instruções depende de uma ação, de um cálculo, etc.

Estrutura de Repetição **enquanto**

Enquanto

- Nesta estrutura, uma expressão lógica é avaliada.
- Enquanto a mesma for verdadeira, uma lista de comandos será executada.
- A execução da lista **se repetirá** até que a avaliação da condição resulte em **FALSO**.

Sintaxe

```
enquanto (condição) {
```

```
// Lista de instruções
```

```
}
```

Exemplo 1

Usando a estrutura de repetição enquanto, faça um algoritmo que escreve os números entre 1 e 10 (incluindo-os).

*Se fosse até 100?

Exemplo 1

Resposta

```
programa {  
    funcao inicio() {  
        inteiro numero  
        numero = 1  
        enquanto (numero <= 10) {  
            escreva(numero, " ")  
            numero = numero + 1  
        }  
    }  
}
```


Exemplo 2

Faça um algoritmo que lê diversos números positivos e escreve o dobro de cada um.

Quando um número negativo for digitado, o algoritmo deverá parar de ler números.

Exemplo 2

Resposta

```
programa {  
    funcao inicio() {  
        inteiro numero, dobro  
        escreva("Digite um número:")  
        leia(numero)  
        enquanto (numero >= 0) {  
            dobro = numero * 2  
            escreva (dobro)  
            escreva ("\nDigite um número:")  
            leia(numero)  
        }  
    }  
}
```

Atenção

- Cuidado com laços (loops) infinitos!
- Devemos nos certificar que em algum momento a expressão lógica será **FALSA**!

Exemplo 3 - Algoritmo com laço infinito:

```
programa {  
    funcao inicio() {  
        inteiro numero = 1  
  
        enquanto (numero <= 10) {  
            numero = 1  
            escreva(numero, " ")  
            numero = numero + 1  
        }  
    }  
}
```

ONDE
ESTÁ O
ERRO?



Exercícios

Exercício 1

Faça um algoritmo que receba números inteiros e, **enquanto números positivos forem digitados, verifique se cada número é menor que 5**. Quando isso acontecer, escreva uma mensagem afirmando que o número é menor que 5.

Quando um número negativo for digitado, o programa vai encerrar.

```
Digite um número inteiro:
9
Digite um número inteiro:
6
Digite um número inteiro:
5
Digite um número inteiro:
4
Foi digitado um número menor que 5
Digite um número inteiro:
2
Foi digitado um número menor que 5
Digite um número inteiro:
-1
Fim do programa
Programa finalizado. Tempo de execução: 18723 ms
```

Exercício 1 - Resposta

```
1  programa {
2      funcao inicio() {
3          inteiro num
4          escreva("Digite um número inteiro:\n")
5          leia(num)
6          enquanto (num>0) {
7              se (num<5) {
8                  escreva("Foi digitado um número menor que 5\n")
9              }
10             escreva("Digite um número inteiro:\n")
11             leia(num)
12         }
13         escreva("Fim do programa")
14     }
15 }
```

```
Digite um número inteiro:
9
Digite um número inteiro:
6
Digite um número inteiro:
5
Digite um número inteiro:
4
Foi digitado um número menor que 5
Digite um número inteiro:
2
Foi digitado um número menor que 5
Digite um número inteiro:
-1
Fim do programa
Programa finalizado. Tempo de execução: 18723 ms
```

Exercício 2

Faça um algoritmo que lê a idade de várias pessoas, até que uma idade negativa seja digitada. O algoritmo deverá calcular e exibir a quantidade de pessoas, de acordo com as faixas etárias apresentadas na tabela abaixo:

FAIXA ETÁRIA	IDADE
1ª	<= 15 anos
2ª	De 16 a 30 anos
3ª	> 30 anos

Exercício 2

Resposta

```
1  programa {
2      funcao inicio() {
3          inteiro idade
4          inteiro total1 = 0, total2 = 0, total3 = 0
5          escreva("Digite uma idade:\n")
6          leia(idade)
7
8          enquanto (idade>=0) {
9              se (idade <= 15) {
10                 total1 = total1 + 1
11             } senao se (idade <=30 ){
12                 total2 = total2 + 1
13             } senao {
14                 total3 = total3+1
15             }
16             escreva("Idade registrada\n")
17             leia(idade)
18         }
19         escreva("\n<=15 anos = ",total1)
20         escreva("\nEntre 16 e 30 anos = ", total2)
21         escreva("\n> 30 anos = ", total3)
22         escreva("\nFim do programa")
23     }
24 }
```

Exercício 3

1. Faça um algoritmo que receba vários números inteiros positivos.
2. O algoritmo deverá encerrar quando um número negativo for digitado.
3. O algoritmo deve registrar o maior número digitado e mostrá-lo ao final.

Digite um número

6

Digite um número

8

Digite um número

10

Digite um número

-1

O maior número digitado foi: 10

Fim do programa

Exercício 3 - Resposta

```
1  ∨ programa {
2  ∨   funcao inicio() {
3      inteiro num, maior = 0
4      escreva ("Digite um número\n")
5      leia(num)
6      enquanto(num>=0) {
7          se (num > maior) {
8              maior = num
9          }
10         escreva ("Digite um número\n")
11         leia(num)
12     }
13     escreva("\nO maior número digitado foi: ", maior)
14     escreva("\nFim do programa")
15 }
16 }
```

Digite um número

6

Digite um número

8

Digite um número

10

Digite um número

-1

O maior número digitado foi: 10

Fim do programa

Estrutura de Repetição

faca...enquanto

faca...enquanto

- Nesta estrutura, todas as instruções são executadas e uma expressão lógica é avaliada.
- Sua lista de expressões **se repetirá** enquanto a avaliação de uma condição resulte em **VERDADEIRO**.
- Cada repetição da lista de instruções contida na estrutura é chamada de **iteração**.
- A diferença entre a estrutura de repetição enquanto e a **faca...enquanto**, é que o **loop é executado pelo menos uma vez, mesmo se a condição for falsa** (porque a condição fica no final da estrutura).

Sintaxe

Utilizamos as palavras reservadas **faca** e **enquanto** para formar o laço de repetição.

```
faca {  
    // Lista de instruções  
} enquanto (condição)
```

Exemplo 4

Faça um algoritmo que escreva os números entre 1 e 10, incluindo-os.

Obs.: Utilize a estrutura de repetição **faca .. enquanto**.

Exemplo 4 - Resposta

```
programa {  
    funcao inicio() {  
        inteiro numero  
        numero = 1  
        faca {  
            escreva (numero, " ")  
            numero = numero + 1  
        } enquanto (numero <= 10)  
    }  
}
```


Exemplo 5

1. Faça um algoritmo para ler vários números inteiros.
2. O algoritmo deverá encerrar a leitura quando o número 0 for digitado.
3. Ao final, mostre quantas vezes o número 13 foi digitado.

Obs.: Utilize a estrutura de repetição **faca .. enquanto**.

```
Digite um número
5
Digite um número
6
Digite um número
13
Digitou 13!
Digite um número
13
Digitou 13!
Digite um número
0
Total de vezes que o número 13 apareceu: 2
```

Exemplo 5

Resposta

```
programa {  
    funcao inicio() {  
        inteiro numero, contador  
        contador = 0  
        faca {  
            escreva("Digite um número:")  
            leia(numero)  
            se (numero == 13) {  
                contador = contador + 1  
            }  
        } enquanto (numero != 0)  
        escreva(contador)  
    }  
}
```

Exercícios

Exercício 4

Faça um algoritmo que lê vários números (o algoritmo deverá encerrar a leitura quando o número 0 for digitado) e conta quantos números negativos foram digitados.

Obs.: Utilize a estrutura de repetição **faca .. enquanto**.

```
Digite um número:  
5  
Digite um número:  
-1  
Digite um número:  
-5  
Digite um número:  
-6  
Digite um número:  
0  
Quantidade de números negativos digitados: 3
```

Exercício 4 - Resposta

```
1  programa {
2      funcao inicio() {
3
4          inteiro num, cont = 0
5          faca {
6              escreva("Digite um número: \n")
7              leia(num)
8              se (num<0) {
9                  cont = cont + 1
10             }
11         } enquanto ( num!=0 )
12     escreva("Quantidade de números negativos digitados: "+ cont)
13 }
14 }
```

Exercício 5

Faça um algoritmo que exibe a frase “Café com bolacha é joia” 10 vezes.

Obs.: Utilize a estrutura de repetição **faca .. enquanto**.

```
1 - Café com bolacha é joia!  
2 - Café com bolacha é joia!  
3 - Café com bolacha é joia!  
4 - Café com bolacha é joia!  
5 - Café com bolacha é joia!  
6 - Café com bolacha é joia!  
7 - Café com bolacha é joia!  
8 - Café com bolacha é joia!  
9 - Café com bolacha é joia!  
10 - Café com bolacha é joia!
```

```
Programa finalizado.
```

Exercício 5 - Resposta

```
1  programa {
2      funcao inicio() {
3
4          inteiro contador = 1
5
6      faca {
7          escreva(contador, " - Café com bolacha é joia!\n")
8          contador+=1 //ou contador++ ou contador = contador + 1
9      } enquanto (contador<=10)
10
11 }
```

Estruturas de Repetição

para

para

- Nesta estrutura, uma variável de controle é inicializada com um valor. Para cada iteração, seu valor é comparado com uma condição que define um valor final.
- A execução da lista de instruções **se repetirá** até que a variável de controle seja maior que o valor final.
- Para cada iteração, a variável de controle é incrementada.

Sintaxe

```
para ( inicialização ; condição ; passo ) {
```

```
    // Lista de instruções
```

```
}
```

Exemplo 6

Faça um algoritmo que escreva os números entre 1 e 5 (incluindo-os).

Obs.: Utilize a estrutura de repetição **para**.

Exemplo 6 - Resposta

```
programa {  
    funcao inicio() {  
        inteiro numero  
  
        para (numero = 1; numero <= 5; numero++) {  
            escreva (numero, " ")  
        }  
    }  
}
```

Exemplo 7

Faça um algoritmo que escreva os números pares entre 1 e 10.

Obs.: Utilize a estrutura de repetição **para**.

Exemplo 7 - Resposta

```
programa {  
    funcao inicio() {  
        inteiro numero  
  
        para (numero = 2; numero < 10; numero=numero+2) {  
            escreva (numero, " ")  
        }  
    }  
}
```

Exercícios

Exercício 6

Utilizando a estrutura PARA, faça um algoritmo que apresente todos os números ímpares entre 0 a 100.

Exercício 6 - Resposta

```
1  programa {  
2      funcao inicio() {  
3          inteiro n  
4          para (n=1; n<=100; n+=2) {  
5              escreva(n, " ")  
6          }  
7      }  
8  }  
9
```

.....

1 3 5 7 9 11 13 15 17 19 21 23 25 27 29 31 33 35 37 39 41 43 45 47 49 51 53 55 57 59 61 63 65 67 69 71
73 75 77 79 81 83 85 87 89 91 93 95 97 99

Programa finalizado. Tempo de execução: 80 ms

Exercício 7

Faça um algoritmo que lê o sexo de um grupo de 10 pessoas e calcula:

- A quantidade total de homens.
- A quantidade total de mulheres.

```
Digite o sexo (m ou f)f
Digite o sexo (m ou f)f
Digite o sexo (m ou f)m
Digite o sexo (m ou f)f
Digite o sexo (m ou f)f

Total de homens: 4
Total de mulheres: 6
```

Exercício 7

Resposta

```
Digite o sexo (m ou f):  
Digite o sexo (m ou f)f  
Digite o sexo (m ou f)f  
Digite o sexo (m ou f)m  
Digite o sexo (m ou f)f  
Digite o sexo (m ou f)f  
  
Total de homens: 4  
Total de mulheres: 6
```

```
1  programa {  
2      funcao inicio() {  
3          caracter sexo  
4          inteiro contm = 0, contf = 0  
5          para (inteiro cont = 0; cont<10; cont++) {  
6              escreva("\nDigite o sexo (m ou f)")  
7              leia(sexo)  
8              se (sexo=="m") {  
9                  contm++  
10             } senao se (sexo=="f") {  
11                 contf++  
12             } senao {  
13                 escreva("\nOpção inválida.")  
14             }  
15         }  
16         escreva("\nTotal de homens: ",contm)  
17         escreva("\nTotal de mulheres: ", contf)  
18     }  
19 }  
20
```

Variação de resposta contando os votos inválidos

```
1  programa {
2      funcao inicio() {
3          caracter sexo
4          inteiro contm = 0, contf = 0, continv = 0
5          para (inteiro cont = 0; cont<10; cont++) {
6              escreva("\nDigite o sexo (m ou f)")
7              leia(sexo)
8              se (sexo=="m") {
9                  contm++
10             } senao se (sexo=="f") {
11                 contf++
12             } senao {
13                 escreva("\nOpção inválida.")
14                 continv++
15             }
16         }
17         escreva("\nTotal de homens: ",contm)
18         escreva("\nTotal de mulheres: ", contf)
19         escreva("\nTotal de opções inválidas: ", continv)
20     }
21 }
```

Exercício 8

Faça um algoritmo que lê a idade de 10 pessoas.

Em seguida, o algoritmo deverá apresentar a quantidade de adolescentes (de 12 a 17 anos).

Exercício 8 - Resposta

```
1  programa {
2      funcao inicio() {
3
4          inteiro idade, cont = 0
5          para (inteiro i=1; i<=10; i++) {
6              escreva("Digite uma idade:\n")
7              leia(idade)
8              se ( (idade >=12) e (idade<=17) ) {
9                  cont++
10             }
11         }
12         escreva("\nO total de adolescentes é: ", cont)
13     }
14 }
15
```

Exercício 9

Faça um algoritmo que receba a idade de 10 pessoas, calcule e mostre a quantidade de pessoas em cada faixa:

FAIXA ETÁRIA	IDADE
1	ATÉ 15 ANOS
2	ENTRE 16 ANOS E 30 ANOS
3	ACIMA DE 30 ANOS

```
Digite a idade
```

```
6
```

```
Digite a idade
```

```
60
```

```
Digite a idade
```

```
80
```

```
Faixa 1 - menor que 15: 5
```

```
Faixa 2 - entre 16 e 30: 2
```

```
Faixa 3 - maior que 30: 3
```

```
Programa finalizado.
```

Exercício 9

Resposta

Digite a idade

6

Digite a idade

60

Digite a idade

80

Faixa 1 - menor que 15: 5

Faixa 2 - entre 16 e 30: 2

Faixa 3 - maior que 30: 3

Programa finalizado.

```
1  programa {
2      funcao inicio() {
3          inteiro i, idade, cont1=0, cont2=0, cont3=0
4          para (i=1; i<=10; i++) {
5              escreva("Digite a idade\n")
6              leia(idade)
7              se (idade > 30){
8                  cont3++
9              } senao se(idade > 15) {
10                 cont2++
11             }
12             senao {
13                 cont1++
14             }
15         }
16         escreva("\nFaixa 1 - menor que 15: ", cont1)
17         escreva("\nFaixa 2 - entre 16 e 30: ", cont2)
18         escreva("\nFaixa 3 - maior que 30: ", cont3)
19     }
20 }
```


Exercício 10

Uma loja utiliza o código **V** para transações à vista e **P** para transações a prazo. Faça um algoritmo que receba o código e o valor de quinze transações, calcule e mostre:

1. A quantidade de compras à vista.
2. A quantidade de compras a prazo.
3. O valor total das compras à vista.
4. O valor total das compras a prazo.
5. O valor total das compras efetuadas.

Exercício 11

Faça um algoritmo que receba 5 números inteiros, calcule e mostre:

- A quantidade de números positivos digitados.
- A soma dos números negativos digitados.

Exercício 11 - Resposta

```
1  programa {
2      funcao inicio() {
3          inteiro i, num, cont = 0, soma = 0
4          para(i=1; i<=5; i++) {
5              escreva("Digite um número: ")
6              leia(num)
7              se (num>0) {
8                  cont++
9              } senao {
10                 soma += num
11             }
12         }
13         escreva("\nTotal de números positivos: ", cont)
14         escreva("\nSoma dos números negativos: ", soma)
15     }
16 }
```

```
Digite um número: 3
Digite um número: 7
Digite um número: 10
Digite um número: -8
Digite um número: -15
```

```
Total de números positivos: 3
Soma dos números negativos: -23
```

Exercício 12

Faça um algoritmo que receba 10 idades, pesos e alturas. Em seguida, calcule e mostre:

1. A média das idades das dez pessoas.
2. A quantidade de pessoas com peso superior a 90 quilos e inferior a 1,50 metro.
3. A porcentagem de pessoas com idade entre 10 e 30 anos.

Exercício 12

Resposta

```
1  programa {
2      funcao inicio() {
3          inteiro idade
4          real peso, altura
5          inteiro somaidade = 0, contpessoa = 0, contidade = 0
6          para (inteiro i=1; i<=3; i++) {
7              escreva("Digite idade, peso e altura:\n")
8              leia(idade, peso, altura)
9              somaidade = somaidade + idade
10             se ( (peso>=90) e (altura <=1.50) ) {
11                 contpessoa = contpessoa + 1
12             }
13             se ( (idade>10) e (idade <30) ) {
14                 contidade = contidade + 1
15             }
16         }
```

```
17         real mediaidade = somaidade / 3
18         real porcentagem = ( (100 * contidade) / 3)
19         escreva("A média das idade é: ", mediaidade)
20         escreva("A quantidade de pessoas com mais de 90kg e menos de 1,50m é: ", contpessoa)
21         escreva ("A porcentagem de pessoas entre 10 e 30 anos é: ", porcentagem )
22     }
23 }
```

Exercício 13

Faça um algoritmo que mostre as tabuadas dos números de 1 a 10.

AULA 4 – ESTRUTURAS DE REPETIÇÃO

@PROF.WALTERTRAVASSOS

WALTER.TRAVASSOS@SECTRAS.EDU.BR